# 指定工程所要使用的Qt模块
QT += gui
QT += core widgets sql

# 指定可执行文件或库的基本文件名，默认为当前目录名
TARGET = dllQLsDBManager

# 指定生成makefile类型
TEMPLATE = lib

# 指定工程配置和编译参数
CONFIG += c++11
CONFIG += object_parallel_to_source

# 指定预定义预处理器符号
DEFINES += DLLQLSDBMANAGER_LIBRARY
DEFINES += QT_DEPRECATED_WARNINGS

# 指定C++实现文件
SOURCES += \
    Dao/UserDao.cpp \
    Pojo/User.cpp \
    Service/UserService.cpp \
    qlsdbconnpool.cpp \
    qlsspecification.cpp

# 指定C++头文件
HEADERS += \
    Dao/UserDao.h \
    Pojo/User.h \
    Service/UserService.h \
    dllQLsDBManager_global.h \
    qlsdbconnpool.h \
    qlsspecification.h

# Default rules for deployment.
unix {
    target.path = /usr/lib
}
!isEmpty(target.path): INSTALLS += target

# 指定release模式编译时的配置
CONFIG(release, debug|release) {
    # 指定可执行文件放置的目录
    DESTDIR = ../../../bin/release
    # 指定目标文件(obj)生成的目录
    OBJECTS_DIR = ../../../lib/release/dllQLsDBManager
    # 指定ui_.h文件生成路径
    UI_DIR = release
    # 指定目标库文件放置的目录
#    DLLDESTDIR = ../../../bin/release
}

# 指定debug模式编译时的配置
CONFIG(debug, debug|release) {
    # 指定可执行文件放置的目录
    DESTDIR = ../../../bin/debug
    # 指定目标文件(obj)生成的目录
    OBJECTS_DIR = ../../../lib/debug/dllQLsDBManager
    # 指定ui_.h文件生成路径
    UI_DIR = debug
    # 指定目标库文件放置的目录
#    DLLDESTDIR = ../../../bin/debug
}
